

```

1  BPM 60
2
3  EVENT 1 GESTE_L_MOV_DETECT
4
5  oscrcv L_incli_front 8877 "/0/incl_front" $L_incli_front
6  oscrcv L_incli_lateral 8877 "/0/incl_lat" $L_incli_lateral
7  oscrcv L_horiz 8877 "/0/mov_horiz" $L_mov_horiz
8  oscrcv L_mov 8877 "/0/still" $L_mov
9
10 oscsend crea_track8 "localhost" :7002 "/crea_track8"
11
12 $nim_map_freq1 := NIM { 0 60, 1 20 "linear" }
13 $nim_map_freq2 := NIM { 0 88, 1 40 "linear" }
14 $nim_map_amp := NIM { 0 -30, 1 -6 "exp_out" }
15
16
17
18 group L_mov
19 {
20   @local $last_L_mov, $last_whenever
21   $last_L_mov := 0
22   crea_track8 L_Mov_bow1 0 0.01 0 Bowed_string amp ($nim_map_amp($L_incli_front)) freq ($nim_map_freq1($L_incli_lateral)) #-> TPan8 pos ($nim_map_pos($L_mov_horiz)) #group ser
23   crea_track8 L_Mov_bow2 1 0.01 0 Bowed_string amp ($nim_map_amp($L_incli_front)) freq ($nim_map_freq2($L_incli_lateral)) #-> TPan8 pos ($nim_map_pos($L_mov_horiz)) #group ser
24   crea_track8 L_Mov_bow1 pause 0.01
25   crea_track8 L_Mov_bow2 pause 0.01
26   whenever ($L_mov==$L_mov)
27   {
28     if ($L_mov != $last_L_mov) {
29       $last_L_mov:= $L_mov
30       if($L_mov==1)
31       {
32         crea_track8 L_Mov_bow1 resume 0.01
33         crea_track8 L_Mov_bow2 resume 0.01
34         crea_track8 L_Mov_bow1 set 01_TPan8 pos ($nim_map_pos($L_mov_horiz))
35         crea_track8 L_Mov_bow2 set 01_TPan8 pos ($nim_map_pos($L_mov_horiz))
36
37         whenever ($L_incli_lateral==$L_incli_lateral)
38         {
39           crea_track8 L_Mov_bow1 set 00_Bowed_string amp ($nim_map_amp($L_incli_front)) freq ($nim_map_freq1($L_incli_lateral))
40           crea_track8 L_Mov_bow2 set 00_Bowed_string amp ($nim_map_amp($L_incli_front)) freq ($nim_map_freq1($L_incli_lateral))
41         }
42       }
43     }else{
44       crea_track8 L_Mov_bow1 pause 0.01
45       crea_track8 L_Mov_bow2 pause 0.01
46     }
47   }
48 }
49 }
50
51
52 10 abort L_mov
53 oscoff L_incli_front
54 oscoff L_incli_lateral
55 oscoff L_horiz
56 oscoff L_mov
57
58 EVENT 1 New_synth
59 ...

```

Example of a dynamic instantiation of 2 synthesizers in SuperCollider and the declaration of the mappings to control it with sensors

- Line 1: Definition of tempo in BPM 60 (1 beat = 1 second)
- Line 3: Creation of a new event with the label "GESTE\_L\_MOV\_DETECT", this label will be reflected in the Max/MSP patch
- Lines 5 to 7: Declaration of input OSC channels; name, port (8877), message prefix and the OSC message assigned to a variable \$...
- Line 10: Declaration of a new OSC send message with name "crea\_track8", host (in this case "localhost"), port (7002) and message prefix ("/crea\_track8") to send information to SuperCollider language to instantiate Synthesizers.
- Lines 11 to 14: Declaration of NIM (New Interpolated Map). All the inputs from sensors and kinect are normalized from 0 to 1. In this case the NIM acts as scale function with an interpolation type ("linear", "exp\_out"). Antescofo NIM have different interpolation types (see Antescofo documentation [http://support.ircam.fr/docs/Antescofo/manuals/Reference/compound\\_curve/index.html#examples-of-interpolation-types](http://support.ircam.fr/docs/Antescofo/manuals/Reference/compound_curve/index.html#examples-of-interpolation-types))
- Lines 18 to 49: Declaration of a Group "L\_mov". This group instantiate 2 new synths in SuperCollider via the OSC message "crea\_track8" and use the sensors data to control it.

- Lines 22-23: Instantiation of 2 Bowed\_string synths connected to a Pan8 ugen (via "#->" message) in the SuperCollider server "server1" (the system works with SuperCollider multi-server option). The "amp (\$nim\_map\_amp(\$L\_incli\_front))" declare that the \$L\_incli\_front variable (that receive the left inclination of hand) will control the amplitude of both synths ("L\_Mov\_bow1" and "L\_Mov\_bow2"). The "freq (\$nim\_map\_freq1(\$L\_incli\_lateral))" will control the frequencies of both synthesizers with the lateral inclination of left hand and "pos (\$nim\_map\_pos(\$L\_mov\_horiz))" will control the position in the space (in an 8 channel circular speakers setup).
- Lines 24-25: pause immediately both synths in SuperCollider (note that pause in SuperCollider means no CPU usage).
- Line 26: The Whenever command is used here to receive the "still" information from the sensors through the "\$L\_mov" variable (line 8), each time the "Whenever" command receive a value and pass the test, it will execute the program (into the brackets).
- Line 28 to 30: Filter data if the previous value is different from the previous one value.
- Line 30: if "\$L\_mov==1", means that the sensor is in movement and the program executes the code into the brackets to resume the synths and change parameters for Pan position.
- Line 37: Declaration of a new "whenever". Means that when there is a lateral inclination of the hand (or any lateral movement) the program executes an amplitude change with the "frontal hand inclination" and the frequencies with the lateral inclination of the hand.
- Line 43: else if there is not movement the synths (L\_Mov\_bow1, L\_Mov\_bow2) will pause.
- Line 52: 10 beats or 10 seconds (because of BPM 60) before the program will abort or stop the group "L\_mov".
- Line 53 to 56: The program stops receive/send osc messages for the specific port and prefix.